
Interaction Design

Interaction Design is the research and development of the ways that humans and 
computers interact.

It includes the research and development of hardware and software interfaces to make 
interacting with a computer more efficient and/or more enjoyable.

The most common hardware devices used to interact with desktop computers is the 
mouse and keyboard.

The mouse and keyboard are typically used to manipulate a desktop interface, mostly 
comprised of windows, icons, buttons, etc., that has become the dominant way to work 
with computers.



Mouse



Mouse

The Processing variables mouseX and mouseY store the x-coordinate and y-coordinate 
of the cursor relative to the origin in the upper-left corner of the display window.

If mouseX and mouseY are used in programs without a draw() or if noLoop() is run in 
setup(), the values will always be 0.

// Circle follows the cursor
void setup() {
  size(100, 100);
  smooth();
  noStroke();
}

void draw() {
  background(126);
  ellipse(mouseX, mouseY, 33, 33);
}



// Invert cursor position to create a secondary response
void setup() {
  size(100, 100);
  noStroke();
  smooth();
}
void draw() {
  float x = mouseX;
  float y = mouseY;
  float ix = width - mouseX; // Inverse X
  float iy = mouseY - height; // Inverse Y
  background(126);
  fill(255, 150);
  ellipse(x, height/2, y, y);
  fill(0, 159);
  ellipse(ix, height/2, iy, iy);
}



The Processing variables pmouseX and pmouseY store the mouse values from the 
previous frame. If the mouse does not move, the values will be the same, but if the 
mouse is moving quickly there can be large differences between the values.

// Draw a line between the current and previous positions
void setup() {
  size(100, 100);
  strokeWeight(8);
  smooth();
}

void draw() {
  background(204);
  line(mouseX, mouseY, pmouseX, pmouseY);
}

Tracking the Mouse Over Frames



Exercise: A Simple Scribble Sketch

Experiment with the following code for a simple scribble sketch by changing the stroke 
colour, or thickness (HINT: Think about using the dist() function.):

// Draw a continuous line between current and previous positions
void setup() {
  size(100, 100);
  strokeWeight(8);
  smooth();
  background(255);
}

void draw() {
  line(mouseX, mouseY, pmouseX, pmouseY);
}



Mouse Buttons

Computer mice and other similar input devices typically have between one and three 
buttons, and Processing can detect when these buttons are pressed.

The mousePressed variable is true if any mouse button is pressed and false if no mouse 
button is pressed. The variable mouseButton is LEFT, CENTER, or RIGHT depending on the 
mouse button most recently pressed.

void setup() {
  size(100, 100);
}

void draw() {
  if (mouseButton == LEFT) { fill(0); }
  else if (mouseButton == RIGHT) { fill(255); }
  else { fill(126); }
  rect(25, 25, 50, 50);
}



Mouse Cursor

The cursor icon used to show the position of the mouse can be hidden with the 
noCursor() function and can be set with the cursor() function to appear as 
different icons.

The options for the cursor() function’s single parameter are ARROW, CROSS, HAND, MOVE, 
TEXT, and WAIT.

If you really want to take control of how the mouse cursor looks, a custom cursor can 
be drawn either as an image or some shapes at the mouse’s location using the 
mouseX and mouseY variables.



Keyboard



Keyboard

Processing registers the most recently pressed key and whether a key is currently 
pressed. The boolean variable keyPressed is true if a key is pressed and is false if 
not. The keyPressed variable remains true while the key is held down and becomes 
false only when the key is released.

void setup() {
  size(100, 100);
  strokeWeight(4);
}
void draw() {
  background(204);
  if (keyPressed == true) { // If the key is pressed,
    line(20, 20, 80, 80); // draw a line
  } else { // Otherwise,
    rect(40, 40, 20, 20); // draw a rectangle
  }
}



Getting the value of the key being pressed

The key variable is of the char data type and stores the value of the most recently 
pressed key. The key variable can store only one value at a time. The most recent key 
pressed will be the only value stored in the variable.

PFont font;

void setup() {
  size(100, 100);
  font = loadFont("ThesisMonoLight-72.vlw");
  textFont(font);
}

void draw() {
  background(0);
  text(key, 28, 75);
}



Controlling a sketch using the keyboard 

To determine whether a specific key is currently being pressed we need to combine 
two tests, one for whether any key is currently being pressed and another to test the 
value of the key.

void setup() {
  size(100, 100);
  strokeWeight(4);
}

void draw() {
  background(204);
  if ((keyPressed == true) && (key == 'A')) {
    line(50, 25, 50, 75);
  } else {
    ellipse(50, 50, 50, 50);
  }
}



Checking for uppercase and lowercase keys

If you want to check for both uppercase and lowercase letters, you have to extend the 
relational expression with a logical OR, the || relational operator, for example:

if ((keyPressed == true) && ((key == 'a') || (key == 'A'))) {
  line(50, 25, 50, 75);
} ...

Alternatively, you could do something different depending on whether shift is being 
held down to produce an uppercase letter:

if (keyPressed == true) {
  if (key == ‘a’) {
    line(50, 25, 50, 75);
  } else if (key == ‘A’) {
    ellipse(50, 50, 50, 50);
  }
}



Coded Keys

Processing can also read the values from other keys including the arrow keys and the 
Alt, Control, Shift, Backspace, Tab, Enter, Return, Escape, and Delete keys.

The expression key==CODED is true if the key is coded and false otherwise. The 
variable keyCode stores the ALT, CONTROL, SHIFT, UP, DOWN, LEFT, and RIGHT keys 
as constants.

Even though not alphanumeric, some keys are included in the ASCII / Unicode 
specification (TAB, BACKSPACE, ENTER, RETURN, ESC, and DELETE) and will not be 
identified as coded by Processing.

For cross-platform projects, note that the Enter key is commonly used on PCs and UNIX and 
the Return key is used on Macintosh. Check for both Enter and Return to make sure your 
program will work for all platforms.



int y = 35;

void setup() {
  size(100, 100);
}

void draw() {
  background(204);
  line(10, 50, 90, 50);
  if (key == CODED) {
    if (keyCode == UP) y = 20;
    if (keyCode == DOWN) y = 50;
  } else {
    y = 35;
  }
  rect(25, y, 50, 30);
}



Events



Events

Functions called event handlers alter the normal flow of a program when an action 
such as a key press or mouse movement takes place.

The code inside an event function is run once each time the corresponding event occurs. 

The mouse event functions are:

  mousePressed()     A mouse button has been pressed 
  mouseReleased()    A mouse button has been released 
  mouseMoved()       The mouse has been moved 
  mouseDragged()     The mouse has been “dragged”

Each key press is registered through two keyboard events:

  keyPressed()       A key has been pressed 
  keyReleased()      A key has been released



void setup() {
  size(100, 100);
  fill(0, 102);
}

void draw() { } // Empty draw() keeps the program running

void mousePressed() {
  rect(mouseX, mouseY, 33, 33);
}



int dragX, dragY, moveX, moveY;
void setup() {
  size(100, 100);
  noStroke();
}

void draw() {
  background(204);
  fill(0);
  ellipse(dragX, dragY, 33, 33); // Black circle
  fill(153);
  ellipse(moveX, moveY, 33, 33); // Gray circle
}

void mouseMoved() {
  // Move gray circle
  moveX = mouseX;
  moveY = mouseY;
}

void mouseDragged() {
  // Move black circle
  dragX = mouseX;
  dragY = mouseY;
}



void setup() {
  size(100, 100);
  noStroke();
  fill(255, 51);
}

void draw() { } // Empty draw() keeps the program running

void keyPressed() {
  int y = key - 32;
  rect(0, y, 100, 4);
}



String letters = "";

void setup() {
  textFont(loadFont("Eureka-24.vlw"));
  stroke(255);
  fill(0);
}

void draw() {
  background(204);
  line(textWidth(letters), 0, textWidth(letters), 100);
  text(letters, 0, 50);
}

void keyPressed() {
  if (key == BACKSPACE) { // Backspace
    if (letters.length() > 0) {
      letters = letters.substring(0, letters.length()-1);
    }
  } else if (textWidth(letters+key) < width) {
    letters = letters+key;
  }
}



int back = 102;

// setup() same as before...

void draw() {
  background(back);
  text(letters, 50, 50);
}

void keyPressed() {
  if ((key == ENTER) || (key == RETURN)) {
    letters = letters.toLowerCase();
    println(letters); // Print to console to see input
    if (letters.equals("black")) back = 0;
    if (letters.equals("gray")) back = 204;
    letters = ""; // Clear the variable
  } else if ((key > 31) && (key != CODED)) {
    // If the key is alphanumeric, add it to the String
    letters = letters + key;
  }
}



The noLoop() function can be used to stop draw() from looping. Running loop() 
resumes a paused sketch. The redraw() function runs the code in draw() one time 
and then halts the execution.

The event functions continue to run when a program is paused with noLoop(). The 
loop() function can be used within these events to continue running the code in 
draw(). The redraw() function can be used inside event handler to update the 
display. It’s helpful when the display needn’t be updated continuously.

Controlling the Flow



void setup() {
  size(100, 100);
  noLoop();
}

void draw() {
  background(204);
  line(mouseX, 0, mouseX, 100);
}

void mousePressed() {
  redraw(); // Run the code in draw one time
}



Time



Seconds, Minutes, Hours

Processing programs can read the value of the computer’s clock.

The current second is read with the second() function, returning 0–59.

The current  minute is read with the minute() function, returning 0–59.  

The current hour is read with the hour() function, returning 0–23.

int s = second(); // Returns values from 0 to 59 
int m = minute(); // Returns values from 0 to 59 
int h = hour();   // Returns values from 0 to 23 
println("The time is " + h + ":" + m + ":" + s);



int lastSecond = second();

void draw() { 
  if (second() != lastSecond) { 
    println(hour() + ":" + minute() + ":" + second()); 
    lastSecond = second(); 
  } 
} 

Telling the Time

14:32:15
14:32:16
14:32:17



Drawing the Time

void setup() { 
  textFont(loadFont("Monaco-14.vlw")); 
  textAlign(CENTER, CENTER);
} 

void draw() { 
  background(0); 
  int s = second(); 
  int m = minute(); 
  int h = hour(); 
  // The nf() function spaces the numbers nicely 
  String t = nf(h,2) + ":" + nf(m,2) + ":" + nf(s,2); 
  text(t, width/2, height/2);
} 



void draw() { 
  translate(width/2, height/2);
  background(0); 
  noStroke(); 
  fill(80); 
  // Angles for sin() and cos() start a 3 o'clock, 
  // subtract HALF_PI to make them start at the top 
  ellipse(0, 0, 80, 80); 
  stroke(255);
  float s = map(second(), 0, 60, 0, TWO_PI) - HALF_PI; 
  float m = map(minute(), 0, 60, 0, TWO_PI) - HALF_PI; 
  float h = map(hour() % 12, 0, 12, 0, TWO_PI) - HALF_PI; 
  line(0, 0, cos(s) * 38, sin(s) * 38); 
  line(0, 0, cos(m) * 30, sin(m) * 30); 
  line(0, 0, cos(h) * 25, sin(h) * 25);
}

Drawing the Time

14:03:26



Milliseconds

Each Processing program counts the time passed since the program started. This time 
is stored in milliseconds (thousandths of a second) and is obtained with the 
millis() function.

2000 milliseconds is 2 seconds
200 milliseconds is 0.2 seconds

Use the millis() function to trigger events (e.g. animations) and calculate the passage 
of time.



Controlling Animations

// Uses millis() to start a line in motion
// three seconds after the program starts 
int x = 0; 

void draw() { 
  if (millis() > 3000) { 
    x++; 
  } 
  line(x, 0, x, 100); 
}



Timing Functions

void drawCircle(float x, float y, int r, int num) {                    
  int startMillis = millis();
  ellipse(x, y, r*2, r*2); 
  if (num > 0) { 
    int branches = int(random(2, 6)); 
    for (int i = 0; i < branches; i++) { 
      float a = random(0, TWO_PI); 
      float newx = x + cos(a) * 6.0 * num; 
      float newy = y + sin(a) * 6.0 * num; 
      drawCircle(newx, newy, r/2, num-1); 
    } 
  } 
  println("drawCircle() took " + millis() - startMillis + "ms");
}


